



プログラミング演習(4)

当たり判定と条件分岐(1)

中村, 青山
小林, 橋本




- Processing で当たり判定に挑戦！
 - 条件分岐を理解する
 - 何らかの条件を満たした時に色を変える！
 - マウスカーソルと動いている円がぶつかったら終了
 - シューティングゲームやもぐらたたきに挑戦！
- 課題：
 - Processing でゲームを作ろう！
 - 占いを作ってみよう

フローチャートと条件分岐

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



・ プログラムの流れ



制御文



```
if( 条件A ){
    // 条件Aの時の処理内容
} else if( 条件B ){
    // 条件Aでなく、条件Bの時の処理内容
} else {
    // 条件AおよびB以外の時の処理内容
}
```

よくあるミス
if(条件);{ 条件A の処理 }

論理値, 真偽値(boolean)



- true か false かを値として持つ
 - それ以外の値は持たない
- 制御文で「条件Aを満たす時」というのは「条件Aが真(true)である」と同じ意味
- 制御文で「条件Aを満たさない時」というのは「条件Aが偽(false)である」と同じ意味
 - $x > y$ の条件をみたす場合, $x > y$ は true, みたさない場合は false
 - $x == y$ は x と y が同じ値の場合 true に, 違う値の場合に false となる

条件の記述方法



演算子	意味	プログラム上
$x > y$	x が y より大きい	左記の時にtrue それ以外でfalse
$x < y$	x が y より小さい	同上
$x \geq y$	x が y 以上	同上
$x \leq y$	x が y 以下	同上
$x == y$	x と y が等しい	同上
$x != y$	x と y が等しくない	同上
$!x$	x は false ?	同上

条件の結果




- 条件の結果はtrue(真)とfalse(偽)になる
- さて、以下の結果はそれぞれ何になる？
 - `println(5 < 3);`
 - `println(5 > 3);`
 - `println(5 == 3);`
 - `println(5 != 3);`
 - `println(5 == 5);`
 - `println(5 != 5);`

アニメーションで最初に戻る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 400x300のウインドウで、左から右へ移動する円が右端に来たら、左端に戻るようにならう。



アニメーションで最初に戻る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(A) 400x300のウインドウで、左から右へ移動する円が右端に来たら、左端に戻るようにならう。


- 円の中央を変数 x を使って $(x, 150)$ とする
- x は最初0にセットし、drawする度に1増やす
- 画面の端に円が来た時の x の値は400
- x が400になつたら x を0にする!!

フローチャートと条件分岐

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



draw() 内の処理



アニメーションで最初に戻る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int x=0;  
  
void setup(){  
    size(400,300);  
}  
  
void draw(){  
    background(255,255,255);  
    fill(255,0,0);  
    x = x + 1;  
    if( x == 400 ){  
        x = 0;  
    }  
    ellipse(x,150,100,100);  
}
```

x == 400 の判定
x が 400 の時に true
となり括弧内が実行される

アニメーションで最初に戻る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int x=0;  
void setup(){  
    size(400,300);  
}  
void draw(){  
    background(255,255,255);  
    fill(255,0,0);  
    x = x + 1;  
    println( x );  
    println( x == 400 );  
    if( x == 400 ){  
        println( "return" );  
        x = 0;  
    }  
    ellipse(x,150,100,100);  
}
```

値の変化を確認
してみましょう

x の値の変化を表示

x == 400 の変化を表示



returnと表示

場所で色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) マウスが400x400のウインドウの左半分にあれば白背景、右半分であれば赤背景にしたい




場所で色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(A) マウスが400x400のウインドウの左半分にあれば白背景、右半分であれば赤背景にしたい

- マウスのX座標を利用(mouseX)
- 右半分はX座標が200より大きいという条件
- $\text{mouseX} > 200$ なら赤に、違えば白にしたい



```
void setup(){
    size(400, 400);
}
void draw(){
    if( )
        background(255, 0, 0);
    } else {
        background(255, 255, 255);
    }
}
```

printlnで確認する！



(A) マウスが400x400のウインドウの左半分にあれば白背景、右半分であれば赤背景にしたい

- 条件の中に入っているのかどうかを判断するため, `println`で `left` や `right` と表示してみる

```
void setup(){
    size(400,400);
}
void draw(){
    if(
        background(255,0,0);
        println( "right" );
    } else{
        background(255,255,255);
        println( "left" );
    }
}
```

```
P sketch_140513a | Processing 2.1.2
File Edit Sketch Tools Help
Java
sketch_140513a

void setup() {
    size(400, 400);
}
void draw() {
    if( mouseX > 200 ){
        background(255, 0, 0);
        println( "right" );
    } else{
        background(255, 255, 255);
        println( "left" );
    }
}

left
left
left
left
left
left
right
right
right
right
right
right
left
left
left
5
```


フローチャートと条件分岐

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- ・プログラムの流れ

draw() 内の処理





距離で色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 400x300のウインドウの中心から20ピクセルの距離に入ったら赤色に塗りつぶすには？



どんな条件分岐が必要か？

距離で色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 距離の計算は `dist(x1,y1,x2,y2)` ;
- 中心(200,150) マウス位置(mouseX,mouseY)
- 中心からマウスまでの距離は
`dist(200,150,mouseX,mouseY)` ;
- `dist(200,150,mouseX,mouseY)<=20`なら赤色 !

```
void draw(){
    if( dist(200,150,mouseX,mouseY) <= 20 ){
        background(255,0,0);
    } else{
        background(255,255,255);
    }
}
```


setup は省略

距離で色を変える



- ・プログラムの流れと条件分岐


draw() 内の処理



クリックされた場所で色変更



(Q) 400x300のウインドウで半径100の円を描き、円の中でクリックされたら円を赤色、外なら円を白色にするには？



クリックはどう取得する？

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- マウスのクリックは `void mousePressed(){ ... }`
- クリックされた場所は `mousePressed` の{}内の `mouseX`と`mouseY`で取得可能
- (例) ボタンが押されたところに円を描く

```
void setup(){
    size( 400, 300 );
    background( 255, 255, 255 );
}
void draw(){
}
void mousePressed(){
    ellipse( mouseX, mouseY, 10, 10 );
}
```

クリックはどう取得する？

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 下記のようにしてしまうとクリックされた場所に丸が表示されない。なぜ？

```
void setup(){
    size( 400, 300 );
}
void draw(){
    background( 255, 255, 255 );
}
void mousePressed(){
    ellipse( mouseX, mouseY, 10, 10 );
}
```

(答)background で全部を再描画してしまうため！

クリックされた場所で色変更



- マウスのクリックは `void mousePressed(){ ... }`
- クリックされた場所が円の中心(200,150)から、100の距離以内であれば塗り色を赤色に！
 - 100の距離以内だったらフラグを1にする(フラグを立てる)

```
int flag = 0;
void setup(){
    size( 400, 300 );
}
void mousePressed(){
    if( dist(200,150,mouseX,mouseY) < 100 ){
        flag = 1;
    } else {
        flag = 0;
    }
}
```

```
void draw(){
    background( 255, 255, 255 );
    if( flag == 1 ){
        fill( 255, 0, 0 );
    } else {
        fill( 255, 255, 255 );
    }
    ellipse( 200, 150, 200, 200 );
}
```



- 400x300のウインドウの上半分でクリックすると背景を青色、下半分でクリックすると背景を黄色にするプログラムを組んでみましょう
- 400x300のウインドウの右下からの距離が30位内の時に背景を緑色に、それ以外の場合は青色にするプログラムを組んでみましょう
- 400x300のウインドウの中央に半径20ピクセルの円を描画し、マウスカーソルが円の中にある場合は赤色で塗りつぶし、そうでない場合は白色で塗りつぶしてみましょう
 - ヒント `fill` を条件分岐の中に入れる

予習問題

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室






- 400x300のウインドウの左半分でクリックすると「大吉」右半分でクリックすると「大凶」と表示するプログラムを組んでみましょう
 - `text("大吉", X座標, Y座標);`
 - `text("大凶", X座標, Y座標);`
 - を切り替え！

色々な条件に挑戦

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




(Q) 600x400のウインドウでマウスが画面の左で赤背景、中で緑背景、左で青背景にするには？



色々な条件に挑戦




- マウスのX座標 `mouseX` に応じて色を変更
- 左は0~200, 中は200~400, 右は400~600



- もし、`mouseX`が0~200なら赤色, 200~400なら緑色, 400~600なら青色
- 言い換えると, `mouseX < 200`なら赤色, そうでなくて`mouseX < 400`なら緑色, そうでなければ青色

色々な条件に挑戦

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



色々な条件に挑戦

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




```
void setup(){
    size( 600, 400 );
}

void draw(){
    if( mouseX < 200 ){
        background(255,0,0);
    } else if( mouseX < 400 ){
        background(0,255,0);
    } else{
        background(0,0,255);
    }
}
```

予習問題



- 画面を上下に4分割し、1番目(一番上)なら赤色、2番目なら緑色、3番目なら青色、4番目(一番下)なら黒色にしてみましょう



予習問題

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




- 左から右に動いてくる白色の円をマウスでクリックすると赤色で塗りつぶすプログラムに挑戦

円の色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) クリックする度に400x400のウインドウの中心に設置された円の色(直径300)が赤→青→黄→赤→青と変化するプログラムを作る



クリックする度に変化

円の色を変える [解法1]

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 考え方
 - 400x400の中心に直径300の円描画は
 - `size(400,400);` と `ellipse(200,200,300,300);`
 - マウスがクリックされると `mousePressed()` に処理がやってくる
 - 赤→青→黄→赤と変化するということは、マウスのクリック回数が
 - 0回, 3回, 6回, 9回の場合は赤色 `fill(255, 0, 0);`
 - 1回, 4回, 7回, 10回の場合は青色 `fill(0, 0, 255);`
 - 2回, 5回, 8回, 11回の場合は黄色 `fill(255, 255, 0);`
 - クリック回数を数える変数(`click`)を用意し、`click`を3で割った余りが0なら赤、1なら青、2なら黄となる

円の色を変える [解法1]

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int click = 0;  
void setup() {  
    size(400, 400);  
}
```

click 回数を数える変数を用意

```
void draw() {  
    background(255);  
    if ( click % 3 == 0 ) {  
        fill( 255, 0, 0 );  
    } else if ( click % 3 == 1 ) {  
        fill( 0, 0, 255 );  
    } else if ( click % 3 == 2 ) {  
        fill( 255, 255, 0 );  
    }  
    ellipse( 200, 200, 300, 300 );  
}
```

click を 3 で割った余りが 0 の時

click を 3 で割った余りが 1 の時

click を 3 で割った余りが 2 の時

```
void mousePressed() {  
    click = click + 1;  
}
```

クリックの度に click の値を増やす

円の色を変える [解法2]


明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



・考え方

- 整数の変数 flag を用意し, flag によって色を変化させるようにする
 - flag == 0 の場合は赤色 fill(255, 0, 0);
 - flag == 1 の場合は青色 fill(0, 0, 255);
 - flag == 2 の場合は黄色 fill(255, 255, 0);
- マウスがクリックされると mousePressed() に処理がやってくるのでその度に flag の値を増やす
 - ただし, flag == 3 となったら, flag を 0 に戻す

flag



円の色を変える [解法2]

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int flag = 0;
```

```
void setup() {
```

```
    size(400, 400);
```

```
}
```

```
void draw() {
```

```
    background(255);
```

```
    if ( flag == 0 ) {
```

```
        fill( 255, 0, 0 );
```

```
    } else if ( flag == 1 ) {
```

```
        fill( 0, 0, 255 );
```

```
    } else if ( flag == 2 ) {
```

```
        fill( 255, 255, 0 );
```

```
}
```

```
    ellipse( 200, 200, 300, 300 );
```

```
}
```

```
void mousePressed() {
```

```
    flag = flag +
```

```
    if( flag == 3 )
```

```
        flag = 0;
```

```
}
```

```
}
```

色を変更するための変数を用意

flag が 0 の時は赤色

flag が 1 の時は青色

flag が 2 の時は黄色


クリックの度に flag の値を増やし
3になったら0に戻す

円の色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 円内でクリックする度に400x400のウインドウ
の中心に設置された円の色(直径300)が赤→青
→黄→赤→青と変化するプログラムを作る



クリックする度に変化

円の色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



• 考え方

- 赤→青→黄→赤と変化するということは、マウスのクリック回数が
 - 0回, 3回, 6回, 9回の場合は赤色 `fill(255, 0, 0);`
 - 1回, 4回, 7回, 10回の場合は青色 `fill(0, 0, 255);`
 - 2回, 5回, 8回, 11回の場合は黄色 `fill(255, 255, 0);`
- クリック回数を数える変数(`click`)を用意し、`click`を3で割った余りが0なら赤、1なら青、2なら黄となる
- マウスがクリックされると`mousePressed()`に処理がやってくるが、`mouseX`, `mouseY` が円内にあるときだけ `click` の値を増やせば良い！

円の色を変える

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int click = 0;

void setup() {
    size(400, 400);
}

void draw() {
    background(255);
    if ( click % 3 == 0 ) {
        fill( 255, 0, 0 );
    } else if ( click % 3 == 1 ) {
        fill( 0, 0, 255 );
    } else if ( click % 3 == 2 ) {
        fill( 255, 255, 0 );
    }
    ellipse( 200, 200, 300, 300 );
}

void mousePressed() {
    if( dist(200,200,mouseX,mouseY)<150 ){
        click = click + 1;
    }
}
```




円の内部でクリックされた場合に
click の値を増やす

信号を作る課題

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 信号機のプログラムを作ってください. マウスクリックによって光る丸が青→黄→赤→青→...とシフトしていくようにしてください



クリックする度に変化


信号を作る課題

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




・考え方

- 何パターンの描画があるかを整理
 - この場合は3パターン



- それぞれの描画パターンをどう描くかをプログラム
- flag で描画パターンを切り替える(クリックの度に数を増やし, 3になると0にする)

- $\text{flag} == 0$ なら
- $\text{flag} == 1$ なら
- $\text{flag} == 2$ なら



となるように draw 内で条件分岐



```

void setup() {
    size(600, 400);
}

int flag = 0;
void mousePressed() {
    flag = flag + 1;
    if ( flag == 3 ) {
        flag = 0;
    }
}

```

パターンを変更する
ための変数を用意

flag が 0 の時は  に



flag が 1 の時は  に

flag が 2 の時は  に

```

void draw() {
    background(255);
    if ( flag == 0 ) {
        fill( 0, 0, 255 );
        ellipse( 150, 200, 140, 140 );
        fill( 50, 50, 0 );
        ellipse( 300, 200, 140, 140 );
        fill( 50, 0, 0 );
        ellipse( 450, 200, 140, 140 );
    } else if ( flag == 1 ) {
        fill( 0, 0, 50 );
        ellipse( 150, 200, 140, 140 );
        fill( 255, 255, 0 );
        ellipse( 300, 200, 140, 140 );
        fill( 50, 0, 0 );
        ellipse( 450, 200, 140, 140 );
    } else if ( flag == 2 ) {
        fill( 0, 0, 50 );
        ellipse( 150, 200, 140, 140 );
        fill( 50, 50, 0 );
        ellipse( 300, 200, 140, 140 );
        fill( 255, 0, 0 );
        ellipse( 450, 200, 140, 140 );
    }
}

```



演習





- 400x400のウインドウの中心に直径200の円を描き、クリックする度に円の色が「赤→黒→青→黄→白→赤→青→...」と変化するようにせよ
- 上の演習を、円の内部の時のみ実行されるようにしてみよ

端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 400x300のウインドウ内で、画面中央から毎フレーム2ピクセルずつ右側に移動する直径が50の円が右端に来ると跳ね返るようにするには？



端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 考え方
 - 円の座標は (x, 150) なので `ellipse(x, 150, 50, 50);`
 - `draw()` の度に x 座標を 2 ずつ増やす
 - 円がウインドウの右端に接するときの条件を考える
 - 「円の中心座標 + 円の半径 \geq ウインドウの右端」の場合に円がウインドウの右端と接している
 - `if(x+25 >= 400) { 右端に接した時の処理をここに書く }`
 - 円がウインドウの右端に接した時に速度を反転
 - 右端に到着した時に速度を -2 にしたら良い

よくある失敗. 右端で吸着してしまう


端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



• 考え方(続き)

- draw() の度に x 座標を 2 ずつまたは -2 ずつ変更するというのを、変数 vx だけ変更するにする！
- 最初は vx を 2 としておき、衝突すると vx = -2 とすれば、常に $x = x + vx;$ で座標を計算可能！
 - 衝突したら $vx = -vx;$ とするだけでもOK!



端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int x = 200;
int vx = 2;

void setup() {
    size( 400, 300 );
}

void draw() {
    background(255);
    x = x + vx;
    if ( x+25 > width ) {
        x = width - 25;
        vx = -vx;
    }
    ellipse( x, height/2, 50, 50 );
}
```

両端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 400x300のウインドウ内で、画面中央から毎フレーム2ピクセルずつ移動する直径が50の円が右端・左端に来ると跳ね返るようにするには？

- 考え方
 - 左端で衝突する時の条件を整理
 - 衝突した時の速度を反転させる

両端で跳ね返る円を描く

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
int x = 200;
int vx = 2;

void setup() {
    size( 400, 300 );
}


void draw() {
    background(255);
    x = x + vx;
    if ( x+25 > width ) {
        x = width - 25;
        vx = -vx;
    } else if( x-25 < 0 ){
        x = 25;
        vx = -vx;
    }
    ellipse( x, 150, 50, 50 );
}
```

カーソルで停止する

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 両端を行き来する円の上にマウスカーソルがある場合、円を停止させ、マウスカーソルが円から外れると動くようにするには？



カーソルで停止する

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 考え方
 - 基本的に両端への移動は先述の通り
 - カーソルが円の中にある場合には移動しない
→ カーソルが円の外にある場合に移動する
 - 円の中心と、マウスカーソルの距離を計算し、マウスカーソルが円の外にある場合に、移動のための計算を行う
 - 円の外にある場合に、円の中心の座標となる変数xを更新する！



```
int x = 200;  
int vx = 2;  
  
void setup() {  
    size( 400, 300 );  
}  
  
void draw() {  
    background(255);  
    if( dist( x, 150, mouseX, mouseY ) > 25 ){  
        x = x + vx;  
    }  
    if ( x+25 > 400 ) {  
        x = 400 - 25;  
        vx = -vx;  
    } else if( x-25 < 0 ){  
        x = 25;  
        vx = -vx;  
    }  
    ellipse( x, 150, 50, 50 );  
}
```

マウスカーソルが円の
外にある場合に移動する！
円の座標の変数を変更する！

演習

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- ・円を X 方向のみならず Y 方向にも移動させ、端に来ると跳ね返るようにしましょう
- ・マウスカーソルの下にある場合は停止するようにしましょう
- ・衝突毎に色を変更しましょう！